INDUSTRIAL AUTOMATION ENGINEERING PORTFOLIO

Project 7 — Version 2

Industrial Batch Mixing System

PLC / Simulation Portfolio Project

Corrected Final Version 2 — Complete Ladder Source

1. Project Overview

Project 7 Version 2 develops the original batch-mixing project into a structured PLC-controlled sequential process. The corrected final ladder automatically fills Material A, transfers to Material B filling, performs a timed mixing stage, drains the vessel, records a completed batch, prepares the sequence for the next automatic cycle, and provides hold-to-run manual requests for Valve A, Valve B, the mixer, and the drain valve.

I developed the program for an Allen-Bradley MicroLogix 1100 PLC using RSLogix Micro Starter Lite. The final design uses retained sequence states including Batch_Active B3:0/5, Fill_A_Step B3:0/6, Fill_B_Step B3:0/7, Mix_Step B3:1/0, and Drain_Step B3:1/1. Manual operation is separated into Manual_Valve_A_Request B3:2/0, Manual_Valve_B_Request B3:2/1, Manual_Mixer_Request B3:2/2, and Manual_Drain_Request B3:2/3.

The corrected final ladder contains Rungs 0000–0035 followed by END at Rung 0036. It includes mode-conflict detection, level consistency monitoring, latched fault handling, controlled reset, sequence recovery, Start Permissive, Machine Run Request, System Ready, filling timeout monitoring, state-based automatic sequencing, manual request interlocks, batch counting, and operator status indication.

This project was developed and tested in a PLC training/simulation environment. The documented testing validates PLC program behavior and does not represent physical process commissioning or safety certification.

2. Engineering Objective

- Detect simultaneous Auto and Manual mode selection and block conflicting operation.
- Detect physically inconsistent tank-level switch combinations.
- Monitor E-stop and mixer-overload healthy feedback.
- Latch abnormal conditions and require deliberate operator recovery.
- Clear abandoned automatic sequence states during a valid reset.
- Establish Start_Permissive, Machine_Run_Request, and System_Ready states.
- Start an automatic batch only from a confirmed empty-tank condition.
- Represent Fill A, Fill B, Mix, and Drain as dedicated retained sequence states.
- Supervise both filling stages with timeout timers.
- Provide a repeatable timed mixing interval.
- Prevent conflicting fill, mix, and drain output commands.
- Provide hold-to-run manual requests for Valve A, Valve B, Mixer Motor, and Drain Valve.
- Apply the same common Fault_Active override to automatic and manual output paths.
- Generate a Batch_Complete_Event and count one completed batch.
- Provide Running, Ready, Fault, and audible Alarm indication.

3. Functional Requirements

- Mode_Conflict B3:0/3 shall energize when Auto_Mode I:1/3 and Manual_Mode I:1/4 are simultaneously true.
- Level_Conflict_Fault B3:1/5 shall energize for High Level without Material A Level or Material A Level without Low Level.
- Fault_Active B3:0/4 shall latch from Mode Conflict, unhealthy EStop_OK I:1/5, unhealthy Mixer_OL_OK I:1/6, Level Conflict, Fill Timeout, or referenced Drain Timeout.
- Reset_PB I:1/2 shall clear stored fault and automatic sequence bits only when the implemented reset permissives are healthy.
- Start_Permissive B3:0/1 shall require no master fault, healthy E-stop and mixer-overload feedback, and no mode/level conflict.
- Start_PB I:1/0 shall latch Machine_Run_Request B3:0/0; Stop_PB I:1/1 or loss of Start_Permissive shall unlatch it.
- System_Ready B3:0/2 shall require Machine_Run_Request and Start_Permissive.

- Batch_Active B3:0/5 shall latch in Auto mode only from a confirmed empty-tank condition.
- Fill_A_Step B3:0/6 shall control Valve_A O:2/0 until Material_A_Level I:1/8 is reached.
- T4:0 shall use time base 1.0 and preset 120 and shall latch Fill_Timeout_Fault B3:1/3 if the target level remains absent.
- Material A completion shall transfer the sequence from Fill_A_Step to Fill_B_Step B3:0/7.
- Fill_B_Step shall control Valve_B O:2/1 until High_Level_Switch I:1/9 is reached.
- T4:1 shall use time base 1.0 and preset 120 and shall use the common Fill_Timeout_Fault.
- High Level shall transfer the process from Fill_B_Step to Mix_Step B3:1/0.
- Mixer_Motor O:2/2 shall operate during the valid automatic Mix Step; T4:2 shall use time base 1.0 and preset 10.
- T4:2/DN shall transfer the sequence from Mix_Step to Drain_Step B3:1/1.
- Drain_Valve O:2/3 shall operate during Drain_Step only while Valve A, Valve B, and Mixer Motor are off and no fault is active.
- Batch_Complete_Event B3:1/2 shall energize when Low_Level_Switch I:1/7 clears during Drain Step with no fault.
- C5:0 shall count completed batches with preset 9999 before automatic sequence cleanup.
- Manual_Valve_A_Request B3:2/0 shall require System Ready, Manual Mode, Manual_Valve_A_PB I:1/13, no fault, Valve B off, Drain Valve off, and High Level false.
- Manual_Valve_B_Request B3:2/1 shall require System Ready, Manual Mode, Manual_Valve_B_PB I:1/14, no fault, Valve A off, Drain Valve off, and High Level false.
- Manual_Mixer_Request B3:2/2 shall require System Ready, Manual Mode, Manual_Mixer_PB I:1/15, no fault, Low Level true, Valve A off, Valve B off, and Drain Valve off.
- Manual_Drain_Request B3:2/3 shall require System Ready, Manual Mode, Manual_Drain_PB I:1/12, no fault, Valve A off, Valve B off, and Mixer Motor off.
- Each physical process output shall be controlled from one OTE with automatic and manual request paths combined where shown in the final ladder.

4. System Components

Component	Address	Function
Start Push Button	I:1/0	Machine start command
Stop Push Button	I:1/1	Machine stop command
Reset Push Button	I:1/2	Fault and sequence recovery
Auto Mode	I:1/3	Automatic mode selector
Manual Mode	I:1/4	Manual mode selector
EStop_OK	I:1/5	E-stop healthy feedback
Mixer_OL_OK	I:1/6	Mixer overload healthy feedback
Low Level Switch	I:1/7	Low-level process feedback
Material A Level	I:1/8	Intermediate level feedback
High Level Switch	I:1/9	High-level/full feedback
Manual Drain PB	I:1/12	Hold-to-run manual drain
Manual Valve A PB	I:1/13	Hold-to-run manual Valve A
Manual Valve B PB	I:1/14	Hold-to-run manual Valve B
Manual Mixer PB	I:1/15	Hold-to-run manual mixer
Valve A	O:2/0	Material A inlet command
Valve B	O:2/1	Material B inlet command
Mixer Motor	O:2/2	Mixer command
Drain Valve	O:2/3	Drain command
Running Lamp	O:2/4	Active-batch indication
Ready Lamp	O:2/5	Ready/idle indication
Fault Lamp	O:2/6	Master fault indication
Alarm Horn	O:2/7	Audible fault indication

5. PLC Data and Address Assignment

Address	Symbolic Name	Function
B3:0/0	Machine_Run_Request	Latched operator run request
B3:0/1	Start_Permissive	Machine operating permission
B3:0/2	System_Ready	Machine ready state
B3:0/3	Mode_Conflict	Auto/Manual conflict diagnostic
B3:0/4	Fault_Active	Latched master fault
B3:0/5	Batch_Active	Active automatic batch
B3:0/6	Fill_A_Step	Material A fill state
B3:0/7	Fill_B_Step	Material B fill state
B3:1/0	Mix_Step	Mixing state
B3:1/1	Drain_Step	Drain state
B3:1/2	Batch_Complete_Event	Completed batch event
B3:1/3	Fill_Timeout_Fault	Latched fill timeout
B3:1/4	Drain_Timeout_Fault	Referenced/reserved in final fault logic
B3:1/5	Level_Conflict_Fault	Level consistency diagnostic
B3:1/6	Mixer_Fault	Referenced/reserved in reset logic
B3:2/0	Manual_Valve_A_Request	Manual Valve A request
B3:2/1	Manual_Valve_B_Request	Manual Valve B request
B3:2/2	Manual_Mixer_Request	Manual mixer request
B3:2/3	Manual_Drain_Request	Manual drain request
T4:0	Fill_A_Timer	TON, TB 1.0, PRE 120
T4:1	Fill_B_Timer	TON, TB 1.0, PRE 120
T4:2	Mixing_Timer	TON, TB 1.0, PRE 10
C5:0	Batch_Counter	CTU, PRE 9999

6. Control Strategy

I organized the final program around Machine Validation → Permission → Run Request → System Ready → Automatic Batch State / Manual Request → Physical Output. Automatic production uses retained Batch Active, Fill A, Fill B, Mix, and Drain states, while manual operation uses separate hold-to-run request bits.

Mode Conflict and Level Conflict validate selector and process feedback before operation. Start Permissive and Machine Run Request establish System Ready. In Auto mode, an empty vessel initiates Fill A, then Material A Level transfers to Fill B, High Level transfers to Mix, T4:2/DN transfers to Drain, and Low Level OFF generates Batch Complete Event.

The corrected final ladder also combines manual request bits into the same physical output rungs. Valve A, Valve B, Mixer Motor, and Drain Valve each have one OTE. Automatic and manual paths merge before the output, with a common Fault Active interlock where implemented. This prevents duplicate physical-output coils and keeps troubleshooting centralized.

7. Engineering Design Decisions

Design Decision	Engineering Purpose
Mode Conflict diagnostic	Prevent contradictory mode selection
Level Conflict diagnostic	Detect impossible level-switch combinations
Latched Fault Active	Require deliberate fault recovery

Separate fault and sequence reset	Return interrupted batch to known state
Start Permissive + Run Request + System Ready	Separate machine health, operator intent, and authorization
Empty-tank batch start	Prevent new batch with residual material
Dedicated automatic step bits	Make sequence state explicit and retained
Independent fill timers	Supervise Material A and B filling separately
Common Fill Timeout Fault	Provide one clear filling-failure status
Timed Mix Step	Provide repeatable mixing duration
Single OTE per physical output	Avoid duplicate-output/last-rung-wins behavior
Manual request bits	Separate manual command validation from physical outputs
Manual mutual interlocks	Prevent conflicting fill/mix/drain commands
Low-level requirement for manual mixer	Prevent manual dry mixing
Common Fault Active output override	Remove automatic and manual process outputs during fault
Batch Complete Event before cleanup	Preserve scan-order counting behavior
Running/Ready/Fault/Alarm indication	Improve operator awareness

8. Ladder Logic Description

8.1 Rung 0000 — Mode Conflict Detection

Ladder Logic

```
|----[ XIC I:1/3 ]----[ XIC I:1/4 ]----------------------------( OTE B3:0/3 )----|
       Auto_Mode        Manual_Mode                                Mode_Conflict
```

Rung Description:
Rung 0000 checks the two operating-mode selectors. When Auto_Mode I:1/3 and Manual_Mode I:1/4 are ON at the same time, Mode_Conflict B3:0/3 turns ON. Since this rung uses an OTE, the bit drops back out as soon as the conflicting selection is removed.

Engineering Purpose:
I added this check because the batch system should never treat Auto and Manual as valid at the same time. The Mode_Conflict bit is used by the fault and permissive logic so an invalid selector combination prevents normal operation.

8.2 Rung 0001 — Level Conflict Detection

Ladder Logic

```
     +----[ XIC I:1/9 ]-------[ XIO I:1/8 ]------------------------+
     |     High_Level          Material_A_Level                     |
|----+                                                              +----( OTE B3:1/5 )----|
     |                                                              |      Level_Conflict_Fault
     +----[ XIC I:1/8 ]-------[ XIO I:1/7 ]------------------------+
           Material_A_Level    Low_Level_Switch
```

Rung Description:
Rung 0001 checks the three tank-level switches for combinations that do not make physical sense. One branch detects High_Level I:1/9 without Material_A_Level I:1/8; the other detects Material_A_Level without Low_Level_Switch I:1/7. Either condition turns on Level_Conflict_Fault B3:1/5.

Engineering Purpose:
The three switches represent increasing liquid levels, so their states should follow that physical order. I added this diagnostic to catch a wiring, sensor, or simulation condition that would otherwise give the sequence contradictory level information.

8.3 Rung 0002 — Primary Fault Latch

Ladder Logic

```
     +----[ XIC B3:0/3 ]-----------------------------------------------+
     |     Mode_Conflict                                               |
     +----[ XIO I:1/5 ]------------------------------------------------+
     |     EStop_OK                                                    |
|----+----[ XIO I:1/6 ]------------------------------------------------+----( OTL B3:0/4 )----|
           Mixer_OL_OK                                                       Fault_Active
```

Rung Description:

Rung 0002 latches Fault_Active B3:0/4 from the primary machine conditions. Mode_Conflict B3:0/3 makes the first branch true, while the XIO contacts detect loss of the healthy EStop_OK I:1/5 or Mixer_OL_OK I:1/6 signals. The OTL keeps Fault_Active stored after the initiating condition disappears.

Engineering Purpose:

I wanted E-stop, mixer overload, and mode-selection problems to produce the same master shutdown response. Latching the fault also prevents the process from resuming automatically after a temporary abnormal condition.

8.4 Rung 0003 — Process Fault Latch

Ladder Logic

```
     +----[ XIC B3:1/5 ]----------------------------------------------+
     |     Level_Conflict_Fault                                       |
     +----[ XIC B3:1/3 ]----------------------------------------------+
     |     Fill_Timeout_Fault                                         |
|----+----[ XIC B3:1/4 ]----------------------------------------------+----( OTL B3:0/4 )----|
           Drain_Timeout_Fault                                              Fault_Active
```

Rung Description:

Rung 0003 adds the process-related faults to the same master fault. Level_Conflict_Fault B3:1/5, Fill_Timeout_Fault B3:1/3, or Drain_Timeout_Fault B3:1/4 can latch Fault_Active B3:0/4.

Engineering Purpose:

I kept process diagnostics on a separate rung from the basic machine-health faults, but they still need to stop the batch in the same controlled way. In the final ladder, Drain_Timeout_Fault is referenced here even though no generating rung for that bit is shown.

8.5 Rung 0004 — Fault Reset

Ladder Logic

```
|--[ XIC I:1/2 ]---[ XIC I:1/5 ]---[ XIC I:1/6 ]---[ XIO B3:0/3 ]---[ XIO B3:1/5 ]---+--( OTU B3:0/4 )--|
   Reset_PB         EStop_OK        Mixer_OL_OK     Mode_Conflict    Level_Conflict  |   Fault_Active
                                                                                     +--( OTU B3:1/3 )--|
                                                                                     |   Fill_Timeout_Fault
                                                                                     +--( OTU B3:1/4 )--|
                                                                                     |   Drain_Timeout_Fault
                                                                                     +--( OTU B3:1/6 )--|
                                                                                    Mixer_Fault
```

Rung Description:

Rung 0004 handles the fault reset. Reset_PB I:1/2 must be ON, EStop_OK I:1/5 and Mixer_OL_OK I:1/6 must be healthy, and both Mode_Conflict B3:0/3 and Level_Conflict_Fault B3:1/5 must be OFF. When those conditions are satisfied, the OTU instructions clear Fault_Active B3:0/4, Fill_Timeout_Fault B3:1/3, Drain_Timeout_Fault B3:1/4, and Mixer_Fault B3:1/6.

Engineering Purpose:
I did not make Reset an unconditional clear command. The controller first checks that the main monitored conditions are back to normal, so the reset represents recovery rather than simply masking an active problem. Mixer_Fault is cleared here even though the final ladder does not show a rung that generates it.

8.6 Rung 0005 — Batch Sequence Recovery Reset

Ladder Logic

```
|--[ XIC I:1/2 ]--[ XIC I:1/5 ]--[ XIC I:1/6 ]--[ XIO B3:0/3 ]--[ XIO B3:1/5 ]--+--( OTU B3:0/5 )--|
   Reset_PB        EStop_OK        Mixer_OL_OK      Mode_Conflict     Level_Conflict  |   Batch_Active
                                                                                  +--( OTU B3:0/6 )--|
                                                                                  |   Fill_A_Step
                                                                                  +--( OTU B3:0/7 )--|
                                                                                  |   Fill_B_Step
                                                                                  +--( OTU B3:1/0 )--|
                                                                                  |   Mix_Step
                                                                                  +--( OTU B3:1/1 )--|
                                                                                      Drain_Step
```

Rung Description:
Rung 0005 uses the same healthy reset conditions to clear the retained process states. A valid Reset unlatches Batch_Active B3:0/5, Fill_A_Step B3:0/6, Fill_B_Step B3:0/7, Mix_Step B3:1/0, and Drain_Step B3:1/1.

Engineering Purpose:
If a fault or stop interrupts the batch halfway through, those step bits may still represent an unfinished sequence. Clearing them together returns the process to a known state before another batch is started.

8.7 Rung 0006 — Start Permissive

Ladder Logic

```
|--[ XIO B3:0/4 ]--[ XIC I:1/5 ]--[ XIC I:1/6 ]--[ XIO B3:0/3 ]--[ XIO B3:1/5 ]-----( OTE B3:0/1 )--|
   Fault_Active      EStop_OK         Mixer_OL_OK    Mode_Conflict    Level_Conflict      Start_Permissive
```

Rung Description:
Rung 0006 generates Start_Permissive B3:0/1. Fault_Active B3:0/4 must be OFF, EStop_OK I:1/5 and Mixer_OL_OK I:1/6 must be ON, and both Mode_Conflict B3:0/3 and Level_Conflict_Fault B3:1/5 must be OFF. Because the rung uses an OTE, the permissive follows the current machine conditions.

Engineering Purpose:
I grouped the main conditions required for operation into one bit instead of repeating the same contacts throughout the program. The Start and ready logic can then use Start_Permissive as a single indication that the machine is healthy enough to accept a run request.

8.8 Rung 0007 — Machine Run Request Latch

Ladder Logic

```
|----[ XIC I:1/0 ]----[ XIC B3:0/1 ]----------------------------( OTL B3:0/0 )----|
       Start_PB          Start_Permissive                                  Machine_Run_Request
```

Rung Description:
Rung 0007 latches Machine_Run_Request B3:0/0 when Start_PB I:1/0 is pressed while Start_Permissive B3:0/1 is true. The OTL keeps the request ON after the pushbutton is released.

Engineering Purpose:
The Start pushbutton is momentary, so I store the operator's intention to run in an internal bit. This keeps the run request separate from the batch sequence itself.

8.9 Rung 0008 — Machine Stop / Run Request Unlatch

Ladder Logic

```
     +----[ XIC I:1/1 ]------------------------------------------+
     |     Stop_PB                                               |
|----+                                                           +----( OTU B3:0/0 )----|
     |                                                           |      Machine_Run_Request
     +----[ XIO B3:0/1 ]------------------------------------------+
           Start_Permissive
```

Rung Description:
Rung 0008 removes Machine_Run_Request B3:0/0 when Stop_PB I:1/1 is pressed or Start_Permissive B3:0/1 is lost. Either branch can execute the OTU.

Engineering Purpose:
The stored run request should not survive a stop or a loss of operating permission. Requiring a new Start command after either condition gives the restart behavior a clear operator action.

8.10 Rung 0009 — System Ready

Ladder Logic

```
|----[ XIC B3:0/0 ]-----------[ XIC B3:0/1 ]-----------------------------( OTE B3:0/2 )----|
      Machine_Run_Request      Start_Permissive                            System_Ready
```

Rung Description:
Rung 0009 generates System_Ready B3:0/2 when Machine_Run_Request B3:0/0 and Start_Permissive B3:0/1 are both true. If either condition drops out, System_Ready turns OFF immediately.

Engineering Purpose:
I kept System_Ready separate from Machine_Run_Request because one represents the operator's request and the other represents the PLC's current authorization to operate. That distinction is useful when the machine is stopped by a permissive condition.

8.11 Rung 0010 — Automatic Batch Start

Ladder Logic

```
|-[ XIC B3:0/2 ]--[ XIC I:1/3 ]--[ XIO B3:0/5 ]--[ XIO I:1/7 ]--[ XIO I:1/8 ]--[ XIO I:1/9 ]--( OTL B3:0/5 )-|
   System_Ready     Auto_Mode      Batch_Active     Low_Level     Material_A_Level  High_Level      Batch_Active
```

Rung Description:
Rung 0010 starts a new automatic batch. System_Ready B3:0/2 and Auto_Mode I:1/3 must be true, Batch_Active B3:0/5 must be OFF, and all three level switches must indicate an empty vessel. When those conditions are met, the OTL latches Batch_Active.

Engineering Purpose:
I require an empty-tank condition before starting a new batch so the next recipe cycle does not begin with residual material already in the vessel.

8.12 Rung 0011 — Fill A Step Initialization

Ladder Logic

```
|-[ XIC B3:0/5]-[ XIO B3:0/6]-[ XIO B3:0/7]-[ XIO B3:1/0]-[XIO B3:1/1 ]-[ XIO I:1/7 ]-[ XIO I:1/8 ]-[ XIO I:1/9 ]-(OTL B3:0/6)-
   Batch_Active  Fill_A_Step   Fill_B_Step    Mix_Step      Drain_Step      Low_Level    Material_A_LevelHigh_Level    Fill_A_Step
```

Rung Description:
Rung 0011 initializes Fill_A_Step B3:0/6. Batch_Active must be true, none of the other process-step bits may be active, and the three level switches must still indicate an empty tank. The OTL then stores Fill_A_Step as the current sequence state.

Engineering Purpose:
I use a dedicated retained bit for each process stage instead of trying to infer the active stage only from sensors and outputs. This gives the sequence a clear starting state and prevents two stages from being active at once.

8.13 Rung 0012 — Valve A Output

Ladder Logic

```
    +--[ XIC B3:0/5 ]--[ XIC B3:0/6 ]--[ XIC I:1/3 ]--[ XIO I:1/8 ]--[ XIO B3:0/7 ]--[ XIO B3:1/0 ]--[ XIO B3:1/1 ]-+
    |  Batch_Active     Fill_A_Step      Auto_Mode     Material_A_Level  Fill_B_Step    Mix_Step       Drain_Step   |
|--+                                                                                                              +--[ XIO B3:0/4 ]--( OTE O:2/0 )-|
    |                                                                                                                 Fault_Active       Valve_A
    +--[ XIC B3:2/0 ]----------------------------------------------------------------------------------------------
       Manual_Valve_A_Request
```

Rung Description:
Rung 0012 controls Valve_A O:2/0 from either the automatic Fill A path or Manual_Valve_A_Request B3:2/0. In Automatic Mode, Batch_Active B3:0/5, Fill_A_Step B3:0/6, Auto_Mode I:1/3, and the Material A level/interlock conditions must be satisfied. The manual request forms a parallel path around the automatic sequence conditions. Both paths still pass through XIO Fault_Active B3:0/4 before the output can energize.

Engineering Purpose:
I kept the automatic and manual commands in one Valve A output rung so O:2/0 has a single control point. Manual operation can bypass the automatic Fill A sequence, but it cannot bypass the common master-fault interlock.

8.14 Rung 0013 — Fill A Timeout Timer

Ladder Logic

```
|--[ XIC B3:0/5 ]--[ XIC B3:0/6 ]--[ XIC O:2/0 ]--[ XIO I:1/8 ]------------[ TON T4:0 ]----|
    Batch_Active      Fill_A_Step      Valve_A        Material_A_Level         Fill_A_Timer
                                                                               Time Base: 1.0
                                                                               Preset: 120
```

Rung Description:
Rung 0013 runs Fill_A_Timer T4:0 while Batch_Active B3:0/5 and Fill_A_Step B3:0/6 are true, Valve_A O:2/0 is energized, and Material_A_Level I:1/8 is still OFF. The final ladder uses a 1.0-second time base and a preset of 120.

Engineering Purpose:
This timer supervises how long the Material A fill is allowed to continue without reaching its level switch. It is a process check for a failed valve, empty supply, bad level signal, or another condition that prevents the expected transition.

8.15 Rung 0014 — Fill A Timeout Fault

Ladder Logic

```
|----[ XIC T4:0/DN ]----[ XIC B3:0/6 ]----[ XIO I:1/8 ]--------( OTL B3:1/3 )----|
      Fill_A_Timer_DN    Fill_A_Step       Material_A_Level     Fill_Timeout_Fault
```

Rung Description:

Rung 0014 latches Fill_Timeout_Fault B3:1/3 if T4:0/DN becomes true while Fill_A_Step B3:0/6 is still active and Material_A_Level I:1/8 has not been reached. The OTL preserves the fault after the sequence is stopped.

Engineering Purpose:

I want an excessive Fill A duration to remain visible after shutdown. Latching the common fill-timeout bit gives the fault system a clear condition to act on and keeps the cause available for diagnosis.

8.16 Rung 0015 — Fill A Complete / Fill B Transfer

Ladder Logic

```
|--[ XIC B3:0/5 ]---[ XIC B3:0/6 ]---[ XIC I:1/8 ]---[ XIO B3:0/4 ]----+--( OTU B3:0/6 )--|
   Batch_Active     Fill_A_Step     Material_A_Level  Fault_Active     |   Fill_A_Step
                                                                       +--( OTL B3:0/7 )--|
                                                                           Fill_B_Step
```

Rung Description:

Rung 0015 transfers the sequence from Fill A to Fill B. During an active, fault-free batch, Material_A_Level I:1/8 causes Fill_A_Step B3:0/6 to unlatch and Fill_B_Step B3:0/7 to latch in the same rung.

Engineering Purpose:

I used an explicit state transfer so the end of Fill A and the beginning of Fill B happen together. That avoids relying on output conditions to guess which fill stage should be active next.

8.17 Rung 0016 — Valve B Output

Ladder Logic

```
     +-[ XIC B3:0/5 ]--[ XIC B3:0/7 ]--[ XIC I:1/3 ]--[ XIO I:1/9 ]--[ XIO B3:0/6 ]--[ XIO B3:1/0 ]--[ XIO B3:1/1 ]-+
     |  Batch_Active    Fill_B_Step       Auto_Mode       High_Level     Fill_A_Step       Mix_Step      Drain_Step   |
|----+                                                                                                                +--[ XIO B3:0/4 ]--( OTE O:2/1 )--|
     |                                                                                                                    Fault_Active       Valve_B
     +--[ XIC B3:2/1 ]-------------------------------------------------------------------------------------------------------------+
         Manual_Valve_B_Request
```

Rung Description:

Rung 0016 controls Valve_B O:2/1 from either the automatic Fill B path or Manual_Valve_B_Request B3:2/1. The automatic branch requires Batch_Active B3:0/5, Fill_B_Step B3:0/7, Auto_Mode I:1/3, High_Level I:1/9 still OFF, and the other process steps inactive. The manual request provides the parallel command path. Both branches are blocked when Fault_Active B3:0/4 is true.

Engineering Purpose:

I used one physical Valve B output rung for both operating modes. The manual path is independent of the automatic Fill B step, but the same master-fault condition still removes the output.

8.18 Rung 0017 — Fill B Timeout Timer

Ladder Logic

```
|--[ XIC B3:0/5 ]--[ XIC B3:0/7 ]--[ XIC O:2/1 ]--[ XIO I:1/9 ]--------[ TON T4:1 ]----|
   Batch_Active     Fill_B_Step     Valve_B        High_Level             Fill_B_Timer
                                                                           Time Base: 1.0
                                                                           Preset: 120
```

Rung Description:
Rung 0017 runs Fill_B_Timer T4:1 while Batch_Active B3:0/5 and Fill_B_Step B3:0/7 are true, Valve_B O:2/1 is energized, and High_Level I:1/9 is still OFF. The final ladder uses a 1.0-second time base and a preset of 120.

Engineering Purpose:
Fill B needs its own supervision because it is a separate material-transfer stage. This timer checks that the high-level condition is reached within the allowed fill period.

8.19 Rung 0018 — Fill B Timeout Fault

Ladder Logic

```
|----[ XIC T4:1/DN ]----[ XIC B3:0/7 ]----[ XIO I:1/9 ]--------( OTL B3:1/3 )----|
       Fill_B_Timer_DN    Fill_B_Step       High_Level              Fill_Timeout_Fault
```

Rung Description:
Rung 0018 latches Fill_Timeout_Fault B3:1/3 if T4:1/DN is true while Fill_B_Step B3:0/7 remains active and High_Level I:1/9 is still OFF. The same fill-timeout bit is used for both Fill A and Fill B.

Engineering Purpose:
I used one common fill-timeout fault because either filling stage represents the same general process problem: the expected level was not reached in time. The active step and timer identify which stage caused it.

8.20 Rung 0019 — Fill B Complete / Mix Transfer

Ladder Logic

```
|--[ XIC B3:0/5 ]---[ XIC B3:0/7 ]---[ XIC I:1/9 ]---[ XIO B3:0/4 ]-----+-----( OTU B3:0/7 )--|
   Batch_Active      Fill_B_Step      High_Level        Fault_Active     |         Fill_B_Step
                                                                        +------( OTL B3:1/0 )--|
                                                                                Mix_Step
```

Rung Description:
Rung 0019 transfers the batch from Fill B to mixing. When Batch_Active and Fill_B_Step are true, High_Level I:1/9 is reached, and Fault_Active is OFF, the rung unlatches Fill_B_Step B3:0/7 and latches Mix_Step B3:1/0.

Engineering Purpose:
The high-level switch is the process condition that confirms filling is complete. Using it to perform a direct state transfer gives the mixer a clear start condition and prevents Fill B from remaining active.

8.21 Rung 0020 — Mixer Motor Output

Ladder Logic

```
     +--[ XIC B3:0/5 ]--[ XIC B3:1/0 ]--[ XIC I:1/3 ]--[ XIC I:1/9 ]--[ XIO B3:0/6 ]--[ XIO B3:0/7 ]--[ XIO B3:1/1 ]-+
     |  Batch_Active      Mix_Step        Auto_Mode      High_Level      Fill_A_Step      Fill_B_Step      Drain_Step    |
|--+                                                                                                                    +--[ XIO B3:0/4 ]--( OTE O:2/2 --|
     |                                                                                                                  |   Fault_Active      Mixer_Motor
     +--[ XIC B3:2/2 ]--------------------------------------------------------------------------------------------------+
         Manual_Mixer_Request
```

Rung Description:
Rung 0020 controls Mixer_Motor O:2/2 from the automatic Mix path or Manual_Mixer_Request B3:2/2. In Automatic Mode, the batch must be active, Mix_Step B3:1/0 must be true, High_Level I:1/9 must be confirmed, and the Fill A, Fill B, and Drain steps must be OFF. The manual request is the alternate branch. XIO Fault_Active B3:0/4 is common to the final output path.

Engineering Purpose:
I combined the automatic and manual mixer commands at one output coil. This allows maintenance operation without duplicating O:2/2 logic, while the common fault interlock still shuts the mixer down in either mode.

8.22 Rung 0021 — Mixing Timer

Ladder Logic

```
|--[ XIC B3:0/5 ]--[ XIC B3:1/0 ]--[ XIC O:2/2 ]--[ XIO B3:0/4 ]--------[ TON T4:2 ]----|
   Batch_Active       Mix_Step          Mixer_Motor       Fault_Active            Mixing_Timer
                                                                              Time Base: 1.0
                                                                              Preset: 10
```

Rung Description:
Rung 0021 runs Mixing_Timer T4:2 while Batch_Active B3:0/5 and Mix_Step B3:1/0 are true, Mixer_Motor O:2/2 is energized, and Fault_Active B3:0/4 is OFF. The final ladder uses a 1.0-second time base and a preset of 10.

Engineering Purpose:
The timer gives the mixing stage a repeatable duration. I start timing from the actual mixer output so the 10-second interval represents commanded mixing rather than simply the presence of the Mix step bit.

8.23 Rung 0022 — Mixing Complete / Drain Transfer

Ladder Logic

```
|---[ XIC B3:0/5 ]---[ XIC B3:1/0 ]---[ XIC T4:2/DN ]---[ XIO B3:0/4 ]----+----( OTU B3:1/0 )--|
   Batch_Active      Mix_Step        Mixing_Timer_DN     Fault_Active      |   Mix_Step
                                                                           +--( OTL B3:1/1 )--|
                                                                               Drain_Step
```

Rung Description:
Rung 0022 transfers the sequence from mixing to draining. When T4:2/DN is true during an active, fault-free Mix step, the rung unlatches Mix_Step B3:1/0 and latches Drain_Step B3:1/1.

Engineering Purpose:
Once the programmed mix time is complete, the process needs a definite handoff into draining. The paired OTU/OTL instructions make that transition explicit and keep only one of the two steps active.

8.24 Rung 0023 — Drain Valve Output

Ladder Logic

```
    +--[ XIC B3:0/5 ]--[ XIC B3:1/1 ]--[ XIC I:1/3 ]--[ XIO B3:0/4 ]--[ XIO O:2/0 ]--[ XIO O:2/1 ]--[ XIO O:2/2 ]----+
    |  Batch_Active     Drain_Step       Auto_Mode      Fault_Active     Valve_A       Valve_B        Mixer_Motor     |
|--+                                                                                                                +--( OTE O:2/3 )--|
    |                                                                                                                   Drain_Valve
    +--[ XIC B3:2/3 ]-------------------------------------------------------------------------------------------------+
        Manual_Drain_Request
```

Rung Description:
Rung 0023 controls Drain_Valve O:2/3 from either the automatic Drain step or Manual_Drain_Request B3:2/3. The automatic branch requires Batch_Active B3:0/5, Drain_Step B3:1/1, Auto_Mode I:1/3, no active fault, and Valve_A O:2/0, Valve_B O:2/1, and Mixer_Motor O:2/2 all OFF. The manual request forms the parallel command path to the same output.

Engineering Purpose:
The drain valve has one final output coil for both Automatic and Manual operation. In the automatic sequence, the output-level interlocks prevent draining while either inlet valve or the mixer is running; manual permission is handled in the dedicated Manual_Drain_Request rung.

8.25 Rung 0024 — Batch Complete Event

Ladder Logic

```
|--[ XIC B3:0/5 ]---[ XIC B3:1/1 ]----[ XIO I:1/7 ]---[ XIO B3:0/4 ]------( OTE B3:1/2 )--|
    Batch_Active     Drain_Step       Low_Level_Switch    Fault_Active        Batch_Complete_Event
```

Rung Description:
Rung 0024 generates Batch_Complete_Event B3:1/2 when Batch_Active B3:0/5 and Drain_Step B3:1/1 are true, Low_Level_Switch I:1/7 is OFF, and Fault_Active B3:0/4 is OFF. Because this is an OTE, the event exists only while the completed-drain condition is present.

Engineering Purpose:
I use a separate completion event to mark the point where the process has finished a full batch. That event can be used for counting and cleanup without making the drain output itself responsible for those functions.

8.26 Rung 0025 — Batch Counter

Ladder Logic

```
|----[ XIC B3:1/2 ]---------------------------------------[ CTU C5:0 ]----|
      Batch_Complete_Event                                 Batch_Counter
                                                           Preset: 9999
```

Rung Description:
Rung 0025 uses Batch_Complete_Event B3:1/2 to increment Batch_Counter C5:0. The final ladder uses a counter preset of 9999.

Engineering Purpose:
I count the batch only after the process has completed the drain stage. This keeps the production total tied to finished cycles rather than batch starts or intermediate steps.

8.27 Rung 0026 — Drain Step Reset

Ladder Logic

```
|----[ XIC B3:1/2 ]---------------------------------------( OTU B3:1/1 )----|
       Batch_Complete_Event                                 Drain_Step
```

Rung Description:
Rung 0026 unlatches Drain_Step B3:1/1 when Batch_Complete_Event B3:1/2 is true.

Engineering Purpose:
Drain Step has to be cleared after the completed batch is counted so the sequence does not remain in the final process stage. Keeping this cleanup separate also makes the end-of-batch transition easy to follow.

8.28 Rung 0027 — Batch Active Reset

Ladder Logic

```
|----[ XIC B3:1/2 ]---------------------------------------( OTU B3:0/5 )----|
       Batch_Complete_Event                                 Batch_Active
```

Rung Description:
Rung 0027 unlatches Batch_Active B3:0/5 when Batch_Complete_Event B3:1/2 is true.

Engineering Purpose:
Clearing Batch_Active marks the overall cycle as finished. If System_Ready and Auto Mode remain valid and the tank is still empty, the batch-start logic can then initiate the next cycle.

8.29 Rung 0028 — Running Lamp

Ladder Logic

```
|----[ XIC B3:0/5 ]----[ XIO B3:0/4 ]----------------------( OTE O:2/4 )----|
       Batch_Active      Fault_Active                        Running_Lamp
```

Rung Description:
Rung 0028 controls Running_Lamp O:2/4. The lamp is ON while Batch_Active B3:0/5 is true and Fault_Active B3:0/4 is OFF.

Engineering Purpose:
I wanted the running indication to represent the complete batch cycle rather than any single valve or motor. The lamp therefore stays on from batch start through the active process sequence unless a fault occurs.

8.30 Rung 0029 — Ready Lamp

Ladder Logic

```
|----[ XIC B3:0/2 ]----[ XIO B3:0/5 ]----[ XIO B3:0/4 ]----( OTE O:2/5 )----|
       System_Ready      Batch_Active      Fault_Active        Ready_Lamp
```

Rung Description:
Rung 0029 controls Ready_Lamp O:2/5. System_Ready B3:0/2 must be true, while Batch_Active B3:0/5 and Fault_Active B3:0/4 must both be OFF.

Engineering Purpose:
The Ready lamp identifies the specific condition where the machine is healthy and enabled but not currently processing a batch. This makes Ready and Running distinct operator states.

8.31 Rung 0030 — Fault Lamp

Ladder Logic

```
|----[ XIC B3:0/4 ]----------------------------------------( OTE O:2/6 )----|
       Fault_Active                                           Fault_Lamp
```

Rung Description:
Rung 0030 energizes Fault_Lamp O:2/6 whenever Fault_Active B3:0/4 is true.

Engineering Purpose:
The fault lamp gives the operator a direct visual indication of the master fault state. Since it follows Fault_Active, it stays on until the master fault is cleared.

8.32 Rung 0031 — Alarm Horn

Ladder Logic

```
|----[ XIC B3:0/4 ]----------------------------------------( OTE O:2/7 )----|
       Fault_Active                                          Alarm_Horn
```

Rung Description:
Rung 0031 energizes Alarm_Horn O:2/7 whenever Fault_Active B3:0/4 is true.

Engineering Purpose:
The audible alarm uses the same master fault state as the fault lamp so an abnormal condition is indicated both visually and audibly.

8.33 Rung 0032 — Manual Valve A Request

Ladder Logic

```
|--[ XIC B3:0/2 ]--[ XIC I:1/4 ]--[ XIC I:1/13 ]--[ XIO B3:0/4 ]--[ XIO O:2/1 ]--[ XIO O:2/3 ]--[ XIO I:1/9 ]--( OTE B3:2/0 )--|
   System_Ready      Manual_Mode      Manual_Valve_A_PB   Fault_Active      Valve_B          Drain_Valve       High_Level         Manual_Valve_A_Request
```

Rung Description:
Rung 0032 generates Manual_Valve_A_Request B3:2/0. System_Ready B3:0/2 and Manual_Mode I:1/4 must be true, Manual_Valve_A_PB I:1/13 must be held, Fault_Active B3:0/4 must be OFF, Valve_B O:2/1 and Drain_Valve O:2/3 must both be OFF, and High_Level I:1/9 must be false. Because the rung uses an OTE, the request turns OFF as soon as any of these conditions is lost.

Engineering Purpose:
I added these interlocks so Manual Valve A operation is useful for testing without allowing the operator to create an obvious process conflict. The valve cannot be manually requested while Valve B or the drain valve is active, at high level, or during a master fault.

8.34 Rung 0033 — Manual Valve B Request

Ladder Logic

```
|-[ XIC B3:0/2 ]-[ XIC I:1/4 ]-[ XIC I:1/14 ]-[ XIO B3:0/4 ]-[ XIO O:2/0 ]-[ XIO O:2/3 ]-[ XIO I:1/9 ]-( OTE B3:2/1 )|
   System_Ready     Manual_Mode     Manual_Valve_B_PB  Fault_Active         Valve_A     Drain_Valve    High_Level  Manual_Valve_B_Request
```

Rung Description:
Rung 0033 generates Manual_Valve_B_Request B3:2/1. System_Ready B3:0/2 and Manual_Mode I:1/4 must be true, Manual_Valve_B_PB I:1/14 must be held, Fault_Active B3:0/4 must be OFF, Valve_A O:2/0 and Drain_Valve O:2/3 must be OFF, and High_Level I:1/9 must be false.

Engineering Purpose:
Manual Valve B follows the same guarded approach as Manual Valve A. The request is hold-to-run and cannot be established while the other inlet valve or drain valve is active, the tank is already at high level, or the controller is faulted.

8.35 Rung 0034 — Manual Mixer Request

Ladder Logic

```
|-[ XIC B3:0/2 ]-[ XIC I:1/4 ]-[ XIC I:1/15 ]-[ XIO B3:0/4 ]--[ XIC I:1/7 ]--[ XIO O:2/0 ]--[ XIO O:2/1 ]--[ XIO O:2/3 ]--( OTE B3:2/2 )-|
  System_Ready   Manual_Mode   Manual_Mixer_PB  Fault_Active   Low_Level       Valve_A       Valve_B       Drain_Valve Manual_Mixer_Request
```

Rung Description:
Rung 0034 generates Manual_Mixer_Request B3:2/2. The system must be ready and in Manual Mode, Manual_Mixer_PB I:1/15 must be held, Fault_Active B3:0/4 must be OFF, Low_Level_Switch I:1/7 must be true, and Valve_A O:2/0, Valve_B O:2/1, and Drain_Valve O:2/3 must all be OFF.

Engineering Purpose:
I require a minimum liquid level before allowing manual mixer operation and block the request while filling or draining outputs are active. That gives the operator manual control without removing the basic process interlocks around the mixer.

8.36 Rung 0035 — Manual Drain Request

Ladder Logic

```
|--[ XIC B3:0/2 ]--[ XIC I:1/4 ]--[ XIC I:1/12 ]--[ XIO B3:0/4 ]--[ XIO O:2/0 ]--[ XIO O:2/1 ]--[ XIO O:2/2 ]--( OTE B3:2/3 )--|
   System_Ready      Manual_Mode    Manual_Drain_PB  Fault_Active   Valve_A        Valve_B         Mixer_Motor      Manual_Drain_Request
```

Rung Description:
Rung 0035 generates Manual_Drain_Request B3:2/3. System_Ready B3:0/2 and Manual_Mode I:1/4 must be true, Manual_Drain_PB I:1/12 must be held, Fault_Active B3:0/4 must be OFF, and Valve_A O:2/0, Valve_B O:2/1, and Mixer_Motor O:2/2 must all be OFF.

Engineering Purpose:
The manual drain command is allowed only when the other process outputs that would conflict with draining are stopped. Keeping it hold-to-run also makes the drain action dependent on the operator's continuing command.

8.37 Rung 0036 — END

Ladder Logic

```
|-------------------------------------------------------------------------------------------( END )----|
```

Rung Description:
Rung 0036 contains the END instruction, which marks the end of the final Project 7 ladder file.

Engineering Purpose:
This is the program boundary for the implemented Project 7 logic. No ladder logic in this file executes below the END instruction.

8.38 Actual RSLogix Implementation

The following images are extracted from the corrected five-page PV7 ladder document and are the technical source of truth for this portfolio. If any typed rung representation appears ambiguous, the final RSLogix screenshot governs.




0003
Level_Confilct _Fault
B3:1
5
Fill_Timeout_Fault
B3:1
3
Drain_Timeout_Fault
B3:1
4
Fault_Active
B3:0
L
4
0004
Reset_PB
I:1
2
1762-IQ16
EStop_OK
I:1
5
1762-IQ16
Mixer_OL_OK
I:1
6
1762-IQ16
Mode_Conflict
B3:0
3
Level_Confilct _Fault
B3:1
5
Fault_Active
B3:0
U
4
Fill_Timeout_Fault
B3:1
U
3
Drain_Timeout_Fault
B3:1
U
4
Mixer_Fault
B3:1
U
6
0005
Reset_PB
I:1
2
1762-IQ16
EStop_OK
I:1
5
1762-IQ16
Mixer_OL_OK
I:1
6
1762-IQ16
Mode_Conflict
B3:0
3
Level_Confilct _Fault
B3:1
5
Batch_Active
B3:0
U
5
Fill_A_Step
B3:0
U
6
Fill_B_Step
B3:0
U
7
Mix_Step
B3:1
U
0
Drain_Step
B3:1
U
1
0006
Fault_Active
B3:0
4
EStop_OK
I:1
5
1762-IQ16
Mixer_OL_OK
I:1
6
1762-IQ16
Mode_Conflict
B3:0
3
Level_Confilct _Fault
B3:1
5
Start_Permissive
B3:0
1
0007
Start_PB
I:1
0
1762-IQ16
Start_Permissive
B3:0
1
Machin_Run_Request
B3:0
L
0


0008
Stop_PB
I:1
1
1762-IQ16
Start_Permissive
B3:0
1
Machin_Run_Request
B3:0
U
0
0009
Machin_Run_Request
B3:0
0
Start_Permissive
B3:0
1
System_Ready
B3:0
2
0010
System_Ready
Auto_Mode
Batch_Active
Low_Level_Switch
Material_A_Level
High_Level_Switch
Batch_Active
L
0011
Batch_Active
Fill_A_Step
Fill_B_Step
Mix_Step
Drain_Step
Low_Level_Switch
Material_A_Level
High_Level_Switch
Fill_A_Step
0012
Batch_Active
Fill_A_Step
Auto_Mode
Material_A_Level
Fill_B_Step
Mix_Step
Drain_Step
Fault_Active
Valve_A
O:2
1762-OB32T
Manual_Valve_A_Request
B3:2
0013
Batch_Active
Fill_A_Step
Valve_A
Material_A_Level
Fill_A_Timeout
TON
Timer On Delay
Timer T4:0
Time Base 1.0
Preset 120<
Accum 0<
EN
DN
0014
Fill_A_Timeout
T4:0
DN
Fill_A_Step
Material_A_Level
Fill_Timeout_Fault
B3:1
3
0015
Batch_Active
Fill_A_Step
Material_A_Level
Fault_Active
Fill_A_Step
U
Fill_B_Step
L
0016
Batch_Active
Fill_B_Step
Auto_Mode
High_Level_Switch
Fill_A_Step
Mix_Step
Drain_Step
Fault_Active
Valve_B
O:2
1762-OB32T
Manual_Valve_B_Request
B3:2


0017
Batch_Active B3:0 5
Fill_B_Step B3:0 7
Valve_B O:2 1 1762-OB32T
High_Level_Switch I:1 9 1762-IQ16
Fill_B_Timeout
TON
Timer On Delay
Timer T4:1
Time Base 1.0
Preset 120<
Accum 0<
EN
DN
0018
Fill_B_Timeout T4:1 DN
Fill_B_Step B3:0 7
High_Level_Switch I:1 9 1762-IQ16
Fill_Timeout_Fault B3:1 L 3
0019
Batch_Active B3:0 5
Fill_B_Step B3:0 7
High_Level_Switch I:1 9 1762-IQ16
Fault_Active B3:0 4
Fill_B_Step B3:0 U 7
Mix_Step B3:1 L 0
0020
Batch_Active B3:0 5
Mix_Step B3:1 0
Auto_Mode I:1 3 1762-IQ16
High_Level_Switch I:1 9 1762-IQ16
Fill_A_Step B3:0 6
Fill_B_Step B3:0 7
Drain_Step B3:1 1
Fault_Active B3:0 4
Mixer_Motor O:2 2 1762-OB32T
Manual_Mixer_Request B3:2 2
0021
Batch_Active B3:0 5
Mix_Step B3:1 0
Mixer_Motor O:2 2 1762-OB32T
Fault_Active B3:0 4
Mixing_Timer
TON
Timer On Delay
Timer T4:2
Time Base 1.0
Preset 10<
Accum 0<
EN
DN
0022
Batch_Active B3:0 5
Mix_Step B3:1 0
Mixing_Timer T4:2 DN
Fault_Active B3:0 4
Mix_Step B3:1 U 0
Drain_Step B3:1 L 1
0023
Batch_Active B3:0 5
Drain_Step B3:1 1
Auto_Mode I:1 3 1762-IQ16
Fault_Active B3:0 4
Valve_A O:2 0 1762-OB32T
Valve_B O:2 1 1762-OB32T
Mixer_Motor O:2 2 1762-OB32T
Drain_Valve O:2 3 1762-OB32T
Manual_Drain_Request B3:2 3
0024
Batch_Active B3:0 5
Drain_Step B3:1 1
Low_Level_Switch I:1 7 1762-IQ16
Fault_Active B3:0 4
Batch_Complete_Event B3:1 2
0025
Batch_Complete_Event B3:1 2
Batch_Counter
CTU
Count Up
Counter C5:0
Preset 9999<
Accum 2<
CU
DN



Figure P7V2.1 — Corrected final RSLogix ladder screenshot.

9. Sequence of Operation

Step 1 — The PLC checks mode selection, EStop_OK, Mixer_OL_OK, and the level-switch hierarchy.

Step 2 — Healthy conditions establish Start_Permissive; Start_PB latches Machine_Run_Request and establishes System_Ready.

Step 3 — In Auto mode, a confirmed empty tank latches Batch_Active and initializes Fill_A_Step.

Step 4 — Valve A fills until Material_A_Level; T4:0 supervises the maximum fill time.

Step 5 — Material A completion transfers the sequence to Fill_B_Step.

Step 6 — Valve B fills until High_Level_Switch; T4:1 supervises the maximum fill time.

Step 7 — High Level transfers the sequence to Mix_Step; Mixer_Motor runs for the T4:2 ten-second interval.

Step 8 — T4:2/DN transfers the sequence to Drain_Step.

Step 9 — Drain_Valve opens only while the required interlocks are satisfied.

Step 10 — Low_Level_Switch clearing during Drain generates Batch_Complete_Event.

Step 11 — C5:0 counts the completed batch before Drain_Step and Batch_Active are cleared.

Step 12 — If Auto mode and System Ready remain valid and the tank stays empty, the next batch can start.

Step 13 — In Manual mode, the dedicated pushbuttons generate hold-to-run request bits for Valve A, Valve B, Mixer, and Drain only when their corresponding interlocks are satisfied.

Step 14 — The manual request bits feed the same physical output rungs used by the automatic sequence.

10. Testing & Validation

Test	Expected / Verified Function	Status
Machine Start	System enters Ready state	PASS
Fill Material A	Valve A fills to Material A level	PASS
Fill Material B	Valve B fills to High Level	PASS
Mixing Cycle	Mixer operates for programmed duration	PASS
Drain Cycle	Drain valve empties tank	PASS
Batch Completion	Batch counter increments	PASS
Emergency Stop	Machine fault response occurs	PASS
Mixer Overload	Fault generated	PASS
Mode Conflict	Startup inhibited	PASS
Level Conflict	Fault generated	PASS
Fill Timeout	Fault generated	PASS
Manual Valve A	Hold-to-run Valve A request/output	PASS
Manual Valve B	Hold-to-run Valve B request/output	PASS
Manual Mixer	Hold-to-run mixer with level/interlock checks	PASS
Manual Drain	Hold-to-run drain with equipment interlocks	PASS

The supplied testing documentation also references a Drain Timeout test. The corrected final ladder references Drain_Timeout_Fault B3:1/4 but still does not show a rung that generates that fault, so the final portfolio does not claim a screenshot-confirmed Drain Timeout implementation.

Validation is documented as PLC/simulation testing, not physical machine commissioning or safety certification.

11. Results

Project 7 Version 2 now demonstrates both a complete state-based automatic batch sequence and a separate hold-to-run manual maintenance architecture. The corrected final ladder retains the single-output-coil philosophy: each physical actuator has one OTE, while automatic and manual command paths feed that output through controlled logic.

The automatic sequence uses explicit Fill A, Fill B, Mix, and Drain states with two 120-second filling supervision timers and a 10-second Mixing Timer. Manual operation adds independent request validation and interlocks for the same four actuators without duplicating output coils.

12. Challenges & Troubleshooting

Symptom	Diagnostic Path
Machine not ready	B3:0/4 → I:1/5 → I:1/6 → B3:0/3 → B3:1/5 → B3:0/1 → B3:0/0
New batch does not start	B3:0/2 → I:1/3 → empty level inputs → B3:0/5
Valve A automatic failure	B3:0/5 → B3:0/6 → I:1/8 → conflicting states → B3:0/4 → O:2/0
Valve A manual failure	B3:0/2 → I:1/4 → I:1/13 → B3:0/4 → O:2/1/O:2/3/I:1/9 → B3:2/0 → O:2/0
Valve B manual failure	B3:0/2 → I:1/4 → I:1/14 → B3:0/4 → O:2/0/O:2/3/I:1/9 → B3:2/1 → O:2/1
Mixer manual failure	B3:0/2 → I:1/4 → I:1/15 → B3:0/4 → I:1/7 → O:2/0/O:2/1/O:2/3 → B3:2/2 → O:2/2
Drain manual failure	B3:0/2 → I:1/4 → I:1/12 → B3:0/4 → O:2/0/O:2/1/O:2/2 → B3:2/3 → O:2/3
Fill timeout	Output → target level input → T4:0/T4:1 → B3:1/3
Batch count does not increment	B3:1/1 → I:1/7 → B3:1/2 → C5:0

13. Lessons Learned

This project strengthened my understanding of combining state-based automatic sequencing with manual maintenance controls. The key design principle is that manual operation should not bypass the same equipment conflicts that matter in automatic operation.

I also reinforced the value of one physical output coil per actuator. Automatic and manual paths are separated upstream and then merged before a single OTE, making output ownership clear during troubleshooting.

The corrected ladder review also demonstrated why the final PLC screenshot must remain the source of truth: the earlier incomplete upload incorrectly made the manual architecture appear absent.

14. Industrial Relevance

Batch mixing systems commonly require both automatic production and manual maintenance operation. Project 7 demonstrates how a PLC can coordinate filling, mixing, draining, timeout supervision, fault recovery, batch counting, operator indication, and interlocked manual jog commands within one structured program.

A physical production installation could additionally require analog level transmitters, valve position feedback, mixer speed feedback, recipes, process instrumentation, safety-rated hardware, HMI/SCADA integration, data logging, and formal commissioning against actual process times.

15. Version 1 → Version 2 Engineering Progression

Version 1 Foundation	Version 2 Implementation	Engineering Improvement
Basic filling sequence	Batch Active + dedicated step bits	Explicit retained process state
Simple sensor transitions	Stored Fill A, Fill B, Mix, Drain states	Predictable sequence progression
Basic start/stop	Start Permissive + Machine Run Request	Separates permission from command
Limited operating validation	Mode Conflict + Level Conflict	Detects selector and sensor inconsistencies
Limited fault handling	Latched master/process faults	Requires deliberate recovery
Simple reset	Separate fault and sequence reset	Returns interrupted batch to known state
Basic filling	Independent Fill A/Fill B timeout timers	Adds process supervision
Simple mixer timing	Dedicated Mix Step + T4:2	Repeatable mixing state
Basic drain control	Drain interlocked against valves/mixer	Prevents conflicting transfer commands
No structured manual architecture	Four hold-to-run manual request bits	Adds maintenance controls with interlocks
Potential duplicate output control	One OTE per actuator with parallel command paths	Clear physical output ownership
Simple completion	Batch Complete Event + C5:0	Counts completed production only
Limited indication	Running, Ready, Fault, Alarm outputs	Improves operator awareness

16. Project Summary

Project 7 Version 2 — Industrial Batch Mixing System implements a structured automatic and manual batch-control architecture on an Allen-Bradley MicroLogix 1100 PLC. The corrected final ladder contains Rungs 0000–0035 plus END at Rung 0036.

The automatic sequence progresses through Fill Material A → Fill Material B → Mix → Drain → Batch Complete → Count → Cleanup → Automatic Next Batch. Manual Mode provides hold-to-run requests for Valve A, Valve B, Mixer Motor, and Drain Valve with the screenshot-confirmed equipment and level interlocks.

T4:0 and T4:1 use the screenshot-confirmed 1.0-second time base and PRE 120; T4:2 uses 1.0-second time base and PRE 10; C5:0 uses PRE 9999. Drain_Timeout_Fault B3:1/4 and Mixer_Fault B3:1/6 remain referenced/reserved bits where the final screenshot does not show a generation rung.

This corrected portfolio supersedes the previous Project 7 Version 2 document and is based on the complete five-page final ladder.